
State_Variable_OTA_10PF
 *   dsauersanjose@aol.com 
 *   www.idea2ic.com                               R3 = 1/A0
 *   4/15/08             ________________________/\  /\  /\__
 *                      |                          \/  \/    |
 *                      |               R2 = 1/A1            |
 *                      |____________/\  /\  /\_             |
 *                      |              \/  \/   |            |
 *                      |   R1 = 1              |BP          |LP
 *                      |_/\  /\  /\_ HP        |            |
 *          R0 = 1/A0   |   \/  \/   |          |            |
 *     ___              | |\         | XOTA_S1  |  XOTA_S2   |
 *    |IN |_/\  /\  /\__|_|-\        | _______  |  _______   |  ___
 *    |___|   \/  \/  SUMN|  \_______|_|      |_|__|      |__|_|OUT|
 *     _|_                |  /         | 1/s  |    | 1/s  |    |___|
 *    /VIN\             __|+/ OPA1     |______|    |______|
 *    \___/            |  |/
 *      |             _|_        H(s) = 1/(s^2 +s*A1   + A0)
 *     _|_            ///        H(s) = 1/(s^2 +s*w0/Q + w0^2)
 *     ///
 *             Set  A0 = 1 and scale s to 1KHz
 *             Then R2 = Q and  s = 2*PI*1KHz
.OPTIONS  GMIN=1e-12    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN    0      AC     1      DC     0
R0        VIN    SUMN   10k    
R1        SUMN   HP     10k
R2        SUMN   BP     100k
R3        SUMN   LP     10k
XOPA1     SUMN   0      HP    OPA
XOTAS1    HP     BP           OTA_S
XOTAS2    BP     LP           OTA_S
.ac       dec    50     10    10k

*==BiCMOS_OTAs_Can_Use_Internal_Capacitors====

.control
run
plot   db(bp)  db(hp) db(lp)   title   StateVariable_Q_10
alter  R2     resistance =    10k
run
plot  db(bp)  db(hp) db(lp)   title   StateVariable_Q_1
.endc
*          ________________
*        _|_   |  VCC      |
*       /VCC\  |           |   C1
*       \___/   ->       <-    _||_            ___
*        _|_   QP1`|___|'QP2  | ||_|_    _____|OUT|
*        ///    _ '| | |`_    |   ///   _|_   |___|     .SUBCKT   OTA_S  IN   OUT
*              |_____| VC2|___|_______ /   \            ___  _______    ___
*    ___      _| VC1      |_          | Egn2|   ---->  |IN |_|      |__|OUT|
*   |IN |___|'QN1       QN2 `|___    _|  +1 |   ---->  |___| | 1/s  |  |___|
*   |___|   |`->  VE1     <-'|  |   |  \___/                 |______|
*              |__________|    _|_  |____|
*                   |          ///      _|_            s  = jw0    s = 2*PI*Freq
*                  _|_                  ///            @s = jw0    R = 1/sC
*                 / _ \IB
*                 \/ \/
*                 /\_/\        IOUT = IB*TANH( VIN/52mV) =>VIN*IB/52mV
*                 \___/        
*                   |          1e-3 > IB >1e-12  (over production and temp!)
*                  _|_
*                  ///

*   http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html
*
*    ___         O1         C1 10pF   ___     .SUBCKT  OPA INP INN OUT
*   |INP|_     _____        _||______|OUT|    
*   |___| |   _|_   |   R1 | || _|_  |___|     ___   |\
*         |_ /   \  |_/\  _|__ /   \          |INP|__|-\    ___
*    ___    | EGN1|  10K\/  IP| EGN2|  ---->  |___|  |  \__|OUT|
*   |INN|___|  -1 |          _| 90dB|  ---->   ___   |  /  |___|
*   |___|    \___/       IP |  \___/          |INN|__|+/
*              |            |____|            |___|  |/ OPA
*             _|_               _|_
*             ///               /// 

.SUBCKT OPA   INP    INN    OUT



EGN1        O1     0      INP    INN    -1
EGN2        OUT    0      IP     0      -1000000
R1          O1     IP     10k 
C1          OUT    IP     10p
.ends

.SUBCKT OTA_S IN      OUT 
QN1         VC1      IN     VE1    NPNP
QN2         VC2      0      VE1    NPNP 
QP1         VC1      VC1    VCC    PNPP
QP2         VC2      VC1    VCC    PNPP
I1          VE1      0             .52n
VCC         VCC      0             DC     2
EGN2        OUT      0      VC2    0      +1
C1          VC2      0            16p
.ends

.model    NPNP   NPN(   BF=2100 VAF=216 )

.model    PNPP   PNP(   BF=2100 VAF=210)

.end
=====================END_OF_SPICE=======================

Because Bipolar transistors can run at low currents,
audio freq filters can be built completely on 
chip using internal capacitors.

Since the value of Q holds very well,
a 6 pole voltage controlled filter with 
any desired response can be built completely 
on chip.  



However OTA can be used to vary feedback.
So complete on-chip audio filters can be build.
These are in fact voltage controlled filters.
And because feedback is adjustable using OTAs,
a 6 pole butterworth can be electronically 
recongfigured into a 6 pole bessel.


